Normally ordered variables in the q-variant of the 1-co sl_2 algebra.

For pragmatic reasons, $\mathbb{E}[\omega, L, Q, P]$ means $\omega^{-1}(1+\epsilon\,\omega^{-4}\,P)\,\mathrm{Exp}[L+\omega^{-1}\,Q]$, where ω is an ϵ-free scalar, L is linear and contains only c's and b's, Q is a balanced quadratic in the u's and the w's and contains no c's and b's, and P is a balanced quartic polynomial in the c's, u's, and w's.

```
(*The R-matrix and its inverse*)
R⁺[i_,j_] := 𝔼[1, b c_j, u_i w_j, c_i c_j - c_j u_i w_j - 1/4 u_i^2 w_j^2]

R⁻[i_,j_] := 𝔼[1, -b c_j, -t^-1 u_i w_j, -c_i c_j + t^-1 c_i u_i w_j - t^-1 u_i w_j + t^-2/4 u_i^2 w_j^2]
```

```
(*The ribbon element is Drinfeld*t^-1/2
 e^-εc. We assign suitable values to the cups and caps,
following Ohtsuki p.72. Only the left-moving cups and caps need to be recorded.*)
nr[i_] := 𝔼[t^1/2, 0, 0, -t^2 c_i]
nl[i_] := 𝔼[1, 0, 0, 0]
ur[i_] := 𝔼[t^-1/2, 0, 0, t^-2 c_i]
ul[i_] := 𝔼[1, 0, 0, 0]
```

```
(*The q-Logos*)
Λ[k_] := 1/4 (α^2 β^2 + 4 α β δ μ + 2 δ^2 μ^2) (1 - 3 t) (1 - t) - 2 μ^2 (α β + δ μ) c_k t -
  2 β δ μ^2 c_k t u_k + 1/2 β ((α β + 2 δ μ) (1 - 3 t) + 4 α β μ t + 8 δ μ^2 t) u_k +
  1/4 β^2 δ ((1 + μ) (1 - 3 t) + 8 μ t) u_k^2 + 1/2 α (α β + 2 δ μ) (1 - 3 t) w_k - 2 α δ μ^2 c_k t w_k -
  2 δ^2 μ^2 c_k t u_k w_k + (α β + δ μ) (1 - 2 δ^2 (1 - t) t) u_k w_k + 1/2 β δ (1 + μ - 2 δ t + 4 δ μ t) u_k^2 w_k +
  1/4 α^2 δ (1 + μ) (1 - 3 t) w_k^2 + 1/2 α δ (1 + μ - 2 δ t) u_k w_k^2 + 1/4 δ^2 (4 μ + δ^2 (1 - 3 t) (1 - t)) u_k^2 w_k^2
```

```
DP[x_→D_α_, y_→D_β_][P_][f_] :=
 Total[CoefficientRules[P, {x, y}] /. ({m_, n_} → c_) :→ c D[f, {α, m}, {β, n}]]
```

```
CF[𝔼[ω_, L_, Q_, P_]] := Expand /@ Together /@
    𝔼[ω /. b :→ Log[t], L, Q /. b :→ Log[t], P /. b :→ Log[t]];
𝔼 /: 𝔼[ω1_, L1_, Q1_, P1_] ≡ 𝔼[ω2_, L2_, Q2_, P2_] :=
  (ω1 == ω2 ∧ L1 == L2 ∧ Q1 == Q2 ∧ P1 == P2);
𝔼 /: 𝔼[ω1_, L1_, Q1_, P1_] 𝔼[ω2_, L2_, Q2_, P2_] :=
  CF@𝔼[ω1 ω2, L1 + L2, ω2 Q1 + ω1 Q2, ω2^4 P1 + ω1^4 P2];
```

```
N_{u_i_,c_j_,k_}[𝔼[ω_, L_, Q_, P_]] := With[{q = ℯ^-γ β u_k + γ c_k}, CF[
    𝔼[ω, γ c_k + (L /. c_j → 0), ω ℯ^-γ β u_k + (Q /. u_i → 0), ℯ^-q DP_{c_j→D_γ, u_i→D_β}[P][ℯ^q]] /.
     {γ → ∂_{c_j} L, β → ω^-1 ∂_{u_i} Q}]];
N_{w_i_,c_j_,k_}[𝔼[ω_, L_, Q_, P_]] := With[{q = ℯ^γ α w_k + γ c_k}, CF[
    𝔼[ω, γ c_k + (L /. c_j → 0), ω ℯ^γ α w_k + (Q /. w_i → 0), ℯ^-q DP_{c_j→D_γ, w_i→D_α}[P][ℯ^q]] /.
     {γ → ∂_{c_j} L, α → ω^-1 ∂_{w_i} Q}]];
```

```
N_{w_i_,u_j_,k_}[𝔼[ω_, L_, Q_, P_]] := With[{q = (1 - t) μ^-1 α β + μ^-1 β u_k + μ^-1 δ u_k w_k + μ^-1 α w_k}, CF[
    𝔼[μ ω, L, μ ω q + μ (Q /. w_i | u_j → 0), μ^4 ℯ^-q DP_{w_i→D_α, u_j→D_β}[P][ℯ^q] + ω^4 Λ[k]] /. μ →
       1 + (t - 1) δ /. {α → ω^-1 (∂_{w_i} Q /. u_j → 0), β → ω^-1 (∂_{u_j} Q /. w_i → 0), δ → ω^-1 ∂_{w_i,u_j} Q}]];
```

```
m_{i_,j_→k_}[Z_] := Module[{x, y},
  Z // N_{w_i,c_j,x} // N_{w_x,u_j,y} // ReplaceAll[{c_{x|y} → c_x, w_j → w_y}] // N_{u_i,c_x,x} //
    ReplaceAll[{c_{i|x} → c_k, u_{x|y} → u_k, w_y → w_k}] // CF]
```

Testing routines.

```
(*Test meta-associativity*)
Q0 = 𝔼[x, Sum[RandomInteger[{-2, 2}] b c_j, {i, 3}, {j, 3}],
   Sum[RandomInteger[{-2, 2}] u_i w_j, {i, 3}, {j, 3}],
   Sum[RandomInteger[{-2, 2}] u_i w_j, {i, 3}, {j, 3}]];
t1 = Q0 // m_{1,2→1} // m_{1,3→1};
t2 = Q0 // m_{2,3→2} // m_{1,2→1};
t1 ≡ t2
```

True

```
(*Check Reidemeister 2,3*)
(R⁺_{2,1} R⁺_{4,3} R⁺_{6,5} // m_{2,4→x} // m_{1,6→y} // m_{3,5→z}) ≡ (R⁺_{2,1} R⁺_{4,3} R⁺_{6,5} // m_{4,6→x} // m_{2,5→y} // m_{1,3→z})
(R⁻_{1,2} R⁻_{3,4} R⁻_{5,6} // m_{2,4→x} // m_{1,6→y} // m_{3,5→z}) ≡ (R⁻_{1,2} R⁻_{3,4} R⁻_{5,6} // m_{4,6→x} // m_{2,5→y} // m_{1,3→z})
R⁺_{1,2} R⁻_{3,4} // m_{4,2→x} // m_{3,1→y}
```

True

True

𝔼[1, 0, 0, 0]

```
(*rotated crossings*)
t2 = ur_1 R⁻_{2,5} nr_3 ur_4 nr_6 // m_{1,2→1} // m_{1,3→1} // m_{4,5→4} // m_{4,6→4}
t3 = (ul_1 R⁻_{2,5} nl_3 ul_4 nl_6 // m_{1,2→1} // m_{1,3→1} // m_{4,5→4} // m_{4,6→4})
t2 ≡ t3
```

$$\mathbb{E}\left[1,\ -b\,c_4,\ -\frac{u_1 w_4}{t},\ -c_1 c_4 - \frac{u_1 w_4}{t} + \frac{c_1 u_1 w_4}{t} + \frac{u_1^2 w_4^2}{4\,t^2}\right]$$

$$\mathbb{E}\left[1,\ -b\,c_4,\ -\frac{u_1 w_4}{t},\ -c_1 c_4 - \frac{u_1 w_4}{t} + \frac{c_1 u_1 w_4}{t} + \frac{u_1^2 w_4^2}{4\,t^2}\right]$$

True

```
ul₁ nl₂ // m1,2→1
nl₁ ul₂ // m1,2→1
nr₁ ur₂ // m1,2→1
ur₁ nr₂ // m1,2→1
```

$\mathbb{E}[1, 0, 0, 0]$

$\mathbb{E}[1, 0, 0, 0]$

$\mathbb{E}[1, 0, 0, 0]$

$\mathbb{E}[1, 0, 0, 0]$

```
(*the oppositely oriented RII *)
```

$R^{-}_{1,2} R^{+}_{3,4}\, ur_5\, nr_6$ // $m_{1,3\to1}$ // $m_{4,5\to4}$ // $m_{4,2\to4}$ // $m_{4,6\to4}$

$\mathbb{E}[1, 0, 0, 0]$

Computations

Kinks

$Kp[k\_] := \left(R^{+}_{1,2}\, nr_3 \;//\; m_{1,3\to1} \;//\; m_{1,2\to1}\right)$ /. $a\_{}_1 \mapsto a_k$

$R^{+}_{1,2}\, ur_3$ // $m_{2,3\to2}$ // $m_{2,1\to1}$;

$Km[k\_] := \left(R^{-}_{2,1}\, nr_3 \;//\; m_{1,3\to1} \;//\; m_{1,2\to1}\right)$ /. $a\_{}_1 \mapsto a_k$

$R^{-}_{2,1}\, ur_3$ // $m_{2,3\to2}$ // $m_{2,1\to1}$;

Kp[1] Km[2] // $m_{1,2\to1}$

$\mathbb{E}[1, 0, 0, 0]$

```
(*Kinks should be central elements!*)
```

k = $R^{+}_{1,2}\, ur_3$ // $m_{2,3\to2}$ // $m_{2,1\to4}$;

$X = \mathbb{E}\left[1, -b_1 c_1, -\frac{u_1 w_1}{t_1}, -\frac{u_1 w_1}{t_1} + \frac{c_1 u_1 w_1}{t_1} + \frac{u_1^2 w_1^2}{4 t_1^2}\right]$;

(k X // $m_{1,4\to1}$) ≡ (k X // $m_{4,1\to1}$)

True

$R^{+}_{1,2}\, nr_3\, R^{+}_{4,5}\, nr_6$ // $m_{1,3\to1}$ // $m_{1,2\to1}$ // $m_{4,6\to4}$ // $m_{4,5\to4}$ // $m_{1,4\to1}$

```
(*Kinks are inverses*)
```

$R^{+}_{1,2}\, nr_3\, R^{-}_{5,4}\, nr_6$ // $m_{1,3\to1}$ // $m_{1,2\to1}$ // $m_{4,6\to4}$ // $m_{4,5\to4}$ // $m_{1,4\to1}$

$$\mathbb{E}\Big[t, 2 b c_1, u_1 w_1 + \frac{u_1 w_1}{t}, -2 t^4 c_1 + 2 t^4 c_1^2 + 4 t^2 u_1 w_1 + 2 t^3 u_1 w_1 - 4 t^2 c_1 u_1 w_1 - 2 t^3 c_1 u_1 w_1 + \frac{7}{4} u_1^2 w_1^2 + 2 t u_1^2 w_1^2 + \frac{3}{4} t^2 u_1^2 w_1^2\Big]$$

$\mathbb{E}[1, 0, 0, 0]$

```
Kink[n_] :=
  Block[{z},
   If[n == 0, z = 𝔼[1, 0, 0, 0]];
   If[n > 0, z = Product[Kp[j], {j, 1, n}]];
   If[n < 0, z = Product[Km[j], {j, 1, Abs[n]}]];
   Do[z = z // m_{1,k→1}, {k, 2, Abs[n]}]; z = z /. a_1 :→ a_0
  ];
```

Compute Trefoil




```
z31 = R⁺_{1,14} R⁻_{5,2} nr_3 ul_4 R⁺_{19,6} R⁻_{7,10} nl_8 ur_9 R⁺_{11,20} nr_12 ul_13 R⁻_{15,18} nl_16 ur_17;
(Do[z31 = z31 // m_{1,k→1}, {k, 2, 20}]; z31 = z31 /. a_1 :→ a)
FromCoefficientRules[
 CoefficientRules[z31⟦4⟧, {c, u, w}] /. {(e_ → a_) :→ (e → Simplify[a] // Factor)}, {c, u, w}]
```

$$\mathbb{E}\left[-1+\frac{1}{t}+t,\ 0,\ 0,\ 16+\frac{2c}{t^4}-\frac{1}{t^3}-\frac{6c}{t^3}+\frac{4}{t^2}+\frac{10c}{t^2}-\frac{10}{t}-\frac{8c}{t}-18t+8ct+14t^2-10ct^2-7t^3+6ct^3+2t^4-2ct^4+2uw-\frac{2uw}{t^4}+\frac{4uw}{t^3}-\frac{6uw}{t^2}+\frac{2uw}{t}-6tuw+4t^2uw-2t^3uw\right]$$

$$-\frac{2c(-1+t)(1+t)(1-t+t^2)^3}{t^4}+\frac{(-1+t)(1-t+t^2)^2(1-t+2t^2)}{t^3}-\frac{2(1+t)(1-t+t^2)^3uw}{t^4}$$

```
(*Compare with Dror's Trefoil*)
```

$$-\frac{(1-t+t^2)^2(-1+2t-3t^2+2t^3)}{t^3}-\frac{c(1-t+t^2)^3(4+t-5t^2-t^3+t^4)}{2t^4}-\frac{(1-t+t^2)^3(-4-5t+t^3)uw}{2t^4}$$

```
{(-1 + 2 t - 3 t^2 + 2 t^3), (4 + t - 5 t^2 - t^3 + t^4), (-4 - 5 t + t^3)} // Factor
```

$$\{(-1+t)(1-t+2t^2),\ (-1+t)(1+t)(-4-t+t^2),\ (1+t)(-4-t+t^2)\}$$

```
(*Mirror trefoil, *)
```

$z31m = R^{-}_{14,1} R^{+}_{2,5} nr_3 ul_4 R^{-}_{6,19} R^{+}_{10,7} nl_8 ur_9 R^{-}_{20,11} nr_{12} ul_{13} R^{+}_{18,15} nl_{16} ur_{17}$;

$(\text{Do}[z31m = z31m \mathbin{//} m_{1,k\to 1},\ \{k, 2, 20\}];\ z31m = z31m \ /.\ a\_{}_1 \mapsto a)$

FromCoefficientRules[CoefficientRules[z31m[[4]], {c, u, w}] /.
  {(e_ → a_) :→ (e → Simplify[a] // Factor)}, {c, u, w}]

$$\mathbb{E}\left[-1+\frac{1}{t}+t,\ 0,\ 0,\ -16-\frac{2}{t^4}+\frac{2c}{t^4}+\frac{7}{t^3}-\frac{6c}{t^3}-\frac{14}{t^2}+\frac{10c}{t^2}+\frac{18}{t}-\frac{8c}{t}+10t+8ct-4t^2-10ct^2+t^3+6ct^3-2ct^4+2uw-\frac{2uw}{t^4}+\frac{4uw}{t^3}-\frac{6uw}{t^2}+\frac{2uw}{t}-6tuw+4t^2uw-2t^3uw\right]$$

$$-\frac{2c(-1+t)(1+t)(1-t+t^2)^3}{t^4}+\frac{(-1+t)(1-t+t^2)^2(2-t+t^2)}{t^4}-\frac{2(1+t)(1-t+t^2)^3uw}{t^4}$$

```
(*the 1-co invariant DOES distinguish it
  from the usual trefoil. So it is STRONGER than Alexander.*)
(z31m ≡ z31) /. t → 2 // Simplify
```

False

Braid closures

```
(*BraidToR converts the cups and caps of a 1-
 1 tangle that comes from a braid closure. The left-most strands are left open,
while the other strands are closed to the right. It only works for knots,
not links. The conventions for the braid are as in the KnotTheory package.*)
```

```
<< KnotTheory`
BraidToR[Index_, BraidWord_] :=
 Block[{Counter, CurrentStrand, RMatrices, LenBraid, RightCaps, LeftCups},
  LenBraid = Length[BraidWord];
  Counter = 1;
  CurrentStrand = 1;
  RMatrices = Table[If[BraidWord[[k]] > 0, Rp[0, 0], Rm[0, 0]], {k, 1, LenBraid}];
  RightCaps = {};
  LeftCups = {};
  For[i = 1, i ≤ Index, i++,
   For[j = 1, j ≤ LenBraid, j++,
    If[CurrentStrand == Abs[BraidWord[[j]]],
      CurrentStrand++;
      If[BraidWord[[j]] > 0,
       RMatrices[[j]][[1]] = Counter;
       ,
       RMatrices[[j]][[2]] = Counter;
      ];
      Counter++;
      ,

      If[CurrentStrand == Abs[BraidWord[[j]]] + 1,
        CurrentStrand--;
        If[BraidWord[[j]] > 0,
         RMatrices[[j]][[2]] = Counter;,
         RMatrices[[j]][[1]] = Counter;
        ];
        Counter++;

       ];
     ];

   ];

   If[i < Index,
    RightCaps = Append[RightCaps, nr[Counter]];
    Counter++;
    LeftCups = Append[LeftCups, ul[Counter]];
    Counter++;
   ];


  ];
  (Times @@ RMatrices) * (Times @@ RightCaps) * (Times @@ LeftCups)

 ]
```

```
Loading KnotTheory` version of September 6, 2014, 13:37:37.2841.
Read more at http://katlas.org/wiki/KnotTheory.
```

P: Symbol P appears in multiple contexts {KnotTheory`, Global`}; definitions in context KnotTheory` may shadow or be shadowed by other definitions.

Q: Symbol Q appears in multiple contexts {KnotTheory`FastKh`Tangles`, Global`}; definitions in context KnotTheory`FastKh`Tangles` may shadow or be shadowed by other definitions.

```
(*Examples*)
BraidToR[2, {1, 1, -1}]
BraidToR @@ BR[Knot[4, 1]]
```

nr[4] Rm[8, 3] Rp[1, 6] Rp[7, 2] ul[5]

KnotTheory: The minimum braids representing the knots with up to 10 crossings were provided by Thomas Gittings. See arXiv:math.GT/0401051.

nr[4] nr[9] Rm[6, 1] Rm[12, 7] Rp[2, 11] Rp[8, 3] ul[5] ul[10]

```
(*The input of the following program is a braidindex and a braid-
 word with that index. The output is the new 1-co invariant,
zero-framed case. We only list the P/omega^4 part. We list the constant
  coefficient first and then the coefficient of c and last the coefficient
  of uw and divide by omega^3. It looks like the constant term is always
  divisible by omega^2 while the other two are divisible by omega^3,
For reference the last term is omega itself. *)
```

```
InvariantOfBraidClosure[BraidIndex_, BraidWord_] :=
 Block[{z, NumberOfIndices, Writhe, Alex},
  Writhe = Plus @@ (Sign /@ BraidWord);
  NumberOfIndices = 2 Length[BraidWord] + 2 (BraidIndex - 1);
  z = BraidToR[BraidIndex, BraidWord] /.
    {nr[a_] :> nr_a, ul[a_] :> ul_a, Rp[i_, j_] :> R^+_{i,j}, Rm[i_, j_] :> R^-_{i,j}};
  (Do[z = z // m_{1,k→1}, {k, 2, NumberOfIndices}];
   z = (z Kink[-Writhe] // m_{0,1→1}) /. a_1 :> a);
  Alex = z[[1]];
  z = FromCoefficientRules[CoefficientRules[z[[4]], {c, u, w}] /.
     {(e_ → a_) :> (e → Simplify[a] // Factor)}, {c, u, w}];
  Together[{z /. {c → 0, u → 0, w → 0}, Coefficient[z, c],
      Coefficient[z, u w], Alex^4} / Alex^3] // Factor
 ]
```

```
InvariantOfBraidClosure[2, {1, 1, 1}]
```

$$\left\{\frac{(-1+t)\,(1-t+2\,t^2)}{1-t+t^2},\ -\frac{2\,(-1+t)\,(1+t)}{t},\ -\frac{2\,(1+t)}{t},\ \frac{1-t+t^2}{t}\right\}$$

```
InvariantOfBraidClosure[2, {-1, -1, -1}]
```

$$\left\{\frac{(-1+t)\,(2-t+t^2)}{t\,(1-t+t^2)},\ -\frac{2\,(-1+t)\,(1+t)}{t},\ -\frac{2\,(1+t)}{t},\ \frac{1-t+t^2}{t}\right\}$$

```
InvariantOfBraidClosure @@ BR[Knot[4, 1]]
```

$$\left\{-\frac{(-1+t)(1+t)}{t}, \frac{2(-1+t)(1+t)}{t}, \frac{2(1+t)}{t}, -\frac{1-3t+t^2}{t}\right\}$$

```
InvariantOfBraidClosure @@ BR[Knot[5, 1]]
```

$$\left\{\left((-1+t)\left(4-3t+5t^2-3t^3+3t^4-t^5+t^6\right)\right)\Big/\left(t^2\left(1-t+t^2-t^3+t^4\right)\right), -\frac{2(-1+t)(1+t)\left(2-t+2t^2\right)}{t^2}, -\frac{2(1+t)\left(2-t+2t^2\right)}{t^2}, \frac{1-t+t^2-t^3+t^4}{t^2}\right\}$$

```
InvariantOfBraidClosure @@ BR[Knot[5, 2]]
```

$$\left\{-\frac{(-1+t)\left(-9+11t-7t^2+t^3\right)}{t\left(2-3t+2t^2\right)}, -\frac{4(-1+t)(1+t)}{t}, -\frac{4(1+t)}{t}, \frac{2-3t+2t^2}{t}\right\}$$

```
InvariantOfBraidClosure @@ BR[Knot[6, 1]]
```

$$\left\{-\frac{(-1+t)\left(5-11t-t^2+3t^3\right)}{(-2+t)\,t\,(-1+2t)}, \frac{4(-1+t)(1+t)}{t}, \frac{4(1+t)}{t}, -\frac{(-2+t)(-1+2t)}{t}\right\}$$

```
InvariantOfBraidClosure @@ BR[Knot[6, 2]]
```

$$\left\{-\left(\left((-1+t)\left(3-12t+16t^2-12t^3+4t^4-2t^6+t^7\right)\right)\Big/\left(t^2\left(1-3t+3t^2-3t^3+t^4\right)\right)\right), \frac{2(-1+t)(1+t)\left(2-3t+2t^2\right)}{t^2}, \frac{2(1+t)\left(2-3t+2t^2\right)}{t^2}, -\frac{1-3t+3t^2-3t^3+t^4}{t^2}\right\}$$

```
InvariantOfBraidClosure @@ BR[Knot[6, 3]]
```

$$\left\{\frac{(-1+t)(1+t)\left(2-3t+2t^2\right)}{t^2}, -\frac{2(-1+t)(1+t)\left(2-3t+2t^2\right)}{t^2}, -\frac{2(1+t)\left(2-3t+2t^2\right)}{t^2}, \frac{1-3t+5t^2-3t^3+t^4}{t^2}\right\}$$

```
Timing[InvariantOfBraidClosure @@ BR[Knot[10, 111]]]
```

$$\left\{35.4192, \left\{-\left(\left((-1+t)\left(3-9t-11t^2+122t^3-382t^4+758t^5-1074t^6+1122t^7-852t^8+451t^9-147t^{10}+21t^{11}\right)\right)\Big/\left(t^3\left(2-3t+2t^2\right)\left(1-3t+3t^2-3t^3+t^4\right)\right)\right), \frac{1}{t^3}2(-1+t)(1+t)\left(6-18t+23t^2-18t^3+6t^4\right), \frac{1}{t^3}2(1+t)\left(6-18t+23t^2-18t^3+6t^4\right), -\frac{1}{t^3}\left(2-3t+2t^2\right)\left(1-3t+3t^2-3t^3+t^4\right)\right\}\right\}$$

```
(*Let's do a braid-Index 5 knot*)
BR[Knot[10, 115]]
```

KnotTheory: The minimum braids representing the knots with up to 10 crossings were provided by Thomas Gittings. See arXiv:math.GT/0401051.

```
BR[5, {1, -2, 1, 3, 2, 2, -4, -3, 2, -3, -3, -4}]
```

```
Timing[InvariantOfBraidClosure @@ BR[Knot[10, 115]]]
```

$$\left\{34.3642, \left\{-\frac{1}{t^3}(-1+t)(1+t)\left(3-18t+29t^2-18t^3+3t^4\right), \frac{1}{t^3}2(-1+t)(1+t)\left(3-18t+29t^2-18t^3+3t^4\right), \frac{1}{t^3} 2(1+t)\left(3-18t+29t^2-18t^3+3t^4\right), -\frac{1}{t^3}\left(1-9t+26t^2-37t^3+26t^4-9t^5+t^6\right)\right\}\right\}$$

```
(*Some torus knots*)
InvariantOfBraidClosure @@ BR[TorusKnot[4, 5]]
```

$$\left\{\left((-1+t)\left(1-t^3+2t^4+3t^5-3t^7+7t^9+6t^{10}-6t^{11}-5t^{12}+6t^{13}+15t^{14}-13t^{16}+17t^{18}-11t^{21}+12t^{22}\right)\right)\Big/\left(t^5\left(1-t+t^2-t^3+t^4\right)\left(1-t^2+t^4-t^6+t^8\right)\right),\right.$$
$$-\frac{1}{t^6}2(-1+t)(1+t)\left(6-5t+6t^2-5t^3+8t^4-5t^5+8t^6-5t^7+6t^8-5t^9+6t^{10}\right),$$
$$-\frac{1}{t^6}2(1+t)\left(6-5t+6t^2-5t^3+8t^4-5t^5+8t^6-5t^7+6t^8-5t^9+6t^{10}\right),$$
$$\left.\frac{1}{t^6}\left(1-t+t^2-t^3+t^4\right)\left(1-t^2+t^4-t^6+t^8\right)\right\}$$

```
BraidToR @@ BR[5, {1, 4, -3, 1, -3, 2, 4, 4, -2, 3, 1, -2, 4, 1, 1, -2, -2, 3}]
```

```
nr[8] nr[15] nr[23] nr[36] Rm[5, 34] Rm[18, 27] Rm[26, 17] Rm[31, 4] Rm[35, 6]
 Rm[41, 20] Rp[1, 38] Rp[3, 42] Rp[7, 14] Rp[10, 25] Rp[12, 29] Rp[21, 30] Rp[22, 13]
 Rp[28, 11] Rp[33, 44] Rp[39, 2] Rp[40, 19] Rp[43, 32] ul[9] ul[16] ul[24] ul[37]
```

```
InvariantOfBraidClosure @@ BR[5, {1, 4, -3, 1, -3, 2, 4, 4, -2, 3, 1, -2, 4, 1, 1, -2, -2, 3}]
```

$$\left\{\left((-1+t)\left(7-25t+33t^2+32t^3-222t^4+528t^5-810t^6+888t^7-692t^8+377t^9-131t^{10}+25t^{11}\right)\right)\Big/\left(t^4\left(1-t+t^2\right)\left(2-3t+2t^2\right)\right), -\frac{1}{t^4}2(-1+t)(1+t)\left(1-t+t^2\right)^2\left(8-11t+8t^2\right),\right.$$
$$\left.-\frac{2(1+t)\left(1-t+t^2\right)^2\left(8-11t+8t^2\right)}{t^4}, \frac{\left(1-t+t^2\right)^3\left(2-3t+2t^2\right)}{t^4}\right\}$$